I. Stiansdal

1988
nr. 17

help

havforskningsinstituttets egg- og larveprogram

Trond Westgård

A model of the vertical distribution of pelagic eggs. A computer realization.

ISBN 82-7461-003-2

HAVFORSKNINGSINSTITUTTETS EGG- OG LARVEPROGRAM (HELP)

A MODEL OF THE VERTICAL DISTRIBUTION OF PELAGIC EGGS. A COMPUTER REALIZATION.

by

Trond Westgård
Institute of Marine Research
P.O.Box 1870
5024 Bergen, Norway

ABSTRACT

The presented mathematical model and it´s realization in an interactive computer program describes the vertical distribution of pelagic fish and plankton eggs. The water column has a density gradient and the vertical diffusion coefficient is a function of the wind blowing at the surface. The density and turbulence coefficient profiles of the water column are defined to be invariant with time. The model is consequently able to describe only the short term fluctuations of the concentration of eggs. The eggs may be distributed in several size- and buoyancy-groups. The total number of eggs in the water column is constant during one session. Despite the crude formulation of the model it is useful for a number of purposes of which the main are the evaluation of sampling designs for planktonic eggs and the establishment of the coincidence between contaminants and biological resources in a vulnerable life stage.

INTRODUCTION

The present model is mainly the model of the vertical distribution of fish eggs in the homogenous wind mixed upper layers of the ocean as described by Sundby (1983). The present work solves the equations numerically in opposition to the analytical solution that Sundby presented, consequently it has been possible to include a density gradient in the watercolumn. The numerical solution also makes it possible to inspect the transient solution of the equations. This means that it models the time from the eggs are introduced in the watercolumn at certain depths until their vertical concentration profile reach an equilibrium. Possible application areas of the model are :

- Design of efficient sampling strategies of planktonic eggs under different weather, salinity and temperature conditions.

- Estimation of the amount of eggs outside the part of the water column that is sampled.

- Evaluation of the vertical coincidence of pollutants and the spawning products of commercial fish species and important plankton species.

With small modifications the model could also be used for the description of the vertical distribution of oildrops and airbubbles in the water column.

The paper includes an user guide to the computer program. The user may then easily compare the model´s result with observations made in field investigations. Since it is believed that an experimental computer program like the present could easily be modified to solve related problems the computer code is included in Appendix 1.

The present work is a part of the HELP-program at the Institute of Marine Research, Bergen. A research program which main goal is to assess the possible effects of oil spills on vulnerable resources of commercially important fish species along the coast of Norway caused by the increased oil activity in Norwegian waters.

THE GOVERNING EQUATIONS

The vertical distribution of eggs is dependent on the densityprofile of the water column, the vertical eddy turbulence coefficient, the vertical current and the density and diameter of the spheres. When the vertical current is negligible, all horizontal gradients are zero and the vertical physical properties of the water column are supposed to be invariable with time the process is described by the following equations :

$$w(z) = \begin{cases} \frac{1}{18} g d^2 \Delta\varrho(z) \nu(z)^{-1} & \text{when } d < D(z) \quad (1A) \\ k_{11}(d + k_{12}D(z)) |\Delta\varrho(z)|^{\frac{2}{3}} \mathrm{sign}(\Delta\varrho(z)) \; \nu(z)^{-\frac{1}{3}} & \text{when } d > D(z) \quad (1B) \end{cases}$$

$$D(z) = 9^{\frac{1}{3}} g^{-\frac{1}{3}} |\Delta\varrho(z)|^{-\frac{1}{3}} \nu(z)^{\frac{2}{3}} \qquad (2)$$

$$\Delta\varrho(z) = \varrho_s(z) - \varrho_e \qquad (3)$$

$$\varrho_s(z) = \varrho(S(z), T(z), P(z)=0) \qquad (4)$$

$$\nu(z) = k_{51} \exp(k_{52} T(z)) \qquad \text{at } S = 30 \text{ 0/00} \qquad (5)$$

$$K(z=0) = k_{61} + k_{62} W^2 \qquad \text{for } 0 < W < 13 \text{ m/s} \qquad (6)$$

$$\frac{\partial(n(z,t))}{\partial t} = \frac{\partial(w(z)n(z,t))}{\partial z} + \frac{\partial}{\partial z}\left(K(z) \frac{\partial(n(z,t))}{\partial z}\right) + f(z,t) \qquad (7)$$

Equation (7) has the constraints :

$$f(z,t) = 0 \; ; \quad 0 < z < H \; ; \quad \frac{\partial n(0,t)}{\partial z} = \frac{\partial n(H,t)}{\partial z} = 0 \; ; \quad 0 < t$$

The definitions of the variables are :

		Unit
z	= depth (zero at surface positive downwards)	(m)
$w(z)$	= ascending/descending velocity of a sphere at depth z	(ms^{-1})
d	= diameter of a sphere	(m)
$D(z)$	= maximum sphere diameter for Stoke´s equation to apply	(m)
ϱ_e	= density of a sphere	(kgm^{-3})
$\varrho_s(z)$	= density at depth z	(kgm^{-3})
$\nu(z)$	= kinematic molecular viscosity at depth z	(m^2s^{-1})
$S(z)$	= salinity at depth z	(0/00)
$T(z)$	= temperature at depth z	(^0C)

$K(z)$	=	eddy turbulence coefficient at depth z	(m^2s^{-1})
W	=	wind velocity at the surface	(ms^{-1})
$n(z,t)$	=	concentration of spheres at depth z at time t	(m^{-3})
$f(z,t)$	=	birth/death of spheres at depth z at time t	$(m^{-3}s^{-1})$
g	=	acceleration of gravity	(ms^{-2})
k_{ij}	=	necessary constants	
H	=	the bottom depth	(m)

Table 1. Values of necessary constants.

$k_{11} = 8.825.10^{-3}$	$k_{12} = -0.400$
$k_{51} = 1.854.10^{-6}$	$k_{52} = -2.783.10^{-2}$
$k_{61} = 7.610.10^{-3}$	$k_{62} = 2.260.10^{-4}$
$g = 9.81$	

Equation (1) is given in Sundby(1983). Equation (1A) is the well-known Stoke´s equation for the sink/rise velocity of a solid sphere in a continous fluid phase. This equation is valid up to a Reynold number of 0.5 according to Sundby. The intermediate region before the motion becomes completely turbulent is described by equation (1B). The maximum diameter for which the Stoke equation applies given in (2) is derived from (1A) by subsituting w(z) in the left handside of (1A) with :

$$w(z) = \frac{Re\ v(z)}{d} \tag{8}$$

Where Re is Reynold´s number for the motion of the sphere. The critical value of Re is chosen to be 0.5 here and this value is used when (8) is substituted into (1A). Equation (1B) is valid upto Re values of at least 5.0 and consequently cover the actual speed range for planktonic eggs (Sundby,1983).

Equation (4) is the International Equation of State of Sea Water (IES 1980) as given in Fofonoff and Millard, 1983. Since the planktonic eggs is mostly at moderate depths the water is taken to be incompressible, i.e. the secant bulk modulus term in IES 1980 is ignored.

The kinematic molecular viscosity of sea water is a function of the salinity and temperature of which temperature is the most important. Sundby (pers.comm.) using data from Krümmel and Ruppin (1905) arrived at equation (5) which is strictly valid for a salinity of 30 0/00.

The eddy turbulence coefficient given in (6) is valid for the upper wind mixed layers of an homogenous water column. The expression was derived by Sundby (1983) from field data on vertical distribution of fish eggs using the 12 hour mean wind velocity before the sampling of the actual profile.

The last equation (7) is the well-known advection/diffusion equation for a scalar concentration given in standard text books like Pond and Pickard (1983). It should be noted that the advection term has a positive sign and not a negative, this is because w(z) is positive upwards i.e. in the negative z-direction.

Sundby (1983) gives the steady state solution of (7) when K(z) and w(z) are both constant. He also includes the distribution of eggbouyancy and eggdiameter groups, i.e. the distribution of ascending/descending velocities of the eggs, which was anticipated to be a truncated normal distribution.

Here the equations (1)-(7) are assumed to be valid for one specific set of values of the diameter and buoyancy and the computer program allows the user to define a reasonable number of such sets that best fits his actual empirical distribution of egg diameter and buoyancy. It is emphasized that K(z) and w(z) are functions of depth in the present work.

NUMERICAL SOLUTION

Equation (7) is solved by setting up a staggered grid as shown in Fig.1. To ease the implementation of the scheme the concentration of spheres has two dummy grid cells, one just over the surface and one just below the bottom. The content of these cells are always zero. Each of the layers are dZ m thick. The concentration of spheres is measured in the middle of each layer. The vertical velocity and eddy turbulence coefficient are measured between the layers and both are set to zero at the surface and at the bottom to satisfy the constraints of (7) i.e. the numerical equivalents to the solution domain´s boundary conditons.

When K(z) and w(z) are constants the numerical solution of (7) is well described and could be found in for example Ames (1977) and Roache (1972). In our case K(z) and w(z) are functions of depth, z. Since the problem is defined in only one space dimension the time step and grid size could be made small and an explicit scheme has been chosen.

```
                      X          n(iZ=0) = 0

z=0*dZ        ||||||Y||||||      w(iZ=0) = K(iZ=0) = 0

                      X          n(iZ=1)

z=1*dZ        ------Y------      w(iZ=1) ; K(iZ=1)
                      .
                      .
                      X          n(iZ-1)

z=iZ-1*dZ     ------Y------      w(iZ-1)   ; K(iZ-1)

                      X          n(iZ)

z=iZ*dZ       ------Y------      w(iZ) ; K(iZ)
                      .
                      .
                      X          n(nZ)

z=nZ*dZ=H     TTTTTTYTTTTTT      w(nZ) = K(nZ) = 0

                      X          n(nZ+1) = 0
```

Figure 1. The staggered grid used in the computations.

When w is a function of z, the second order upwind differencing method is efficient for the buoyancy flux term term in (7), the method has also been called the donor cell method (Roache, 1972). The numerical representation of the buoyancy term, advn, is then :

$$\text{advn} = \frac{w_1 n_1 - w_2\, n_2}{dZ}\, dT \tag{9}$$

Where

$$w_1 = \frac{1}{2}\,(w(iZ+1)+w(iZ))\ ;\quad w_2 = \frac{1}{2}\,(w(iZ)+w(iZ-1)) \tag{10}$$

$$n_1 = \begin{cases} n(iZ) & \text{for } w_1 > 0 \\ n(iZ+1) & \text{for } w_1 < 0 \end{cases} \tag{11}$$

$$n_2 = \begin{cases} n(iZ-1) & \text{for } w_2 > 0 \\ n(iZ) & \text{for } w_2 < 0 \end{cases} \tag{12}$$

The local change in n(iZ) due to diffusion, difn, is evaluated in a sort of two step Euler scheme (Dag Slagstad, pers.comm.) that is simple and yet efficient.

$$\text{difn} = \frac{K(iZ)(n(iZ+1)-n(iZ)) - K(iZ-1)(n(iZ)-n(iZ-1))}{dZ^2}\, dT \tag{13}$$

It is noticed that when K(z) is constant the scheme degenerates to the usual centered second difference method for the diffusion term.

In the case when w(z) and K(z) are constants Roache op. cit. gives an accurate description of the numerical diffusion for the presented numerical solution of (7). In that case the physical diffusion coefficient could be modified and the steady state solution will exactly match the analytical solution. The error of the diffusion coefficient in the transient solution could also be estimated in that case. In the present paper the physical coefficient of diffusion K(z) has been modified by an approximate numerical diffusion term and this value has been used to modify the K(z) value in each of the nZ layers. For further details the interested reader are referred to the computer code included in the Appendix 1.

A USER GUIDE TO THE PROGRAM

The program is implemented on a ND-500 computer in Fortran-77 and the result is displayed either on a graphical screen or a pen plotter using the GPGS-F library (Anon.,1984). The list of the program is given in Appendix 1.

The user is presented with the following dialogue :

```
Give number of eggdiameter groups : 2
Give number of eggbuoyancy groups : 2
Give number of depth layers       : 50
Give name of file with data       : (your-user-id)your:data
Give maximum depth in meters      : 100.0
Give wind speed in m/s            : 5.0
Give the number of days to run the model : 7.0
Give hours between showtimes             : 0.5
```

The maximum number of egg diameter and egg buoyancy groups is 9 so the maximum number of egg groups is 81. This should be sufficient for most purposes. The maximum number of layers that the user has measured or computed the initial distribution of eggs, salinity, temperature and turbulence in the water column is 200. The datafile is made by the user using an editor.

The layout of the datafile must be as follows :

```
5.0  ┐
5.2  │
 .   │ The temperature in each of the layers (°C)
 .   │
5.0  ┘
34.0 ┐
34.2 │
 .   │ The salinity in each of the layers    (0/00)
 .   │
34.8 ┘
 0.0 ┐
 4.2 │
 .   │ The initial concentration of eggs in each of the layers(no./m3)
 .   │
 0.8 ┘
```

```
1.0  ┐ <-- This number must always be 1.0 !!
1.0  │
 .   │ The relative size of the eddy turbulence in each layer
 .   │ compared to the wind induced value in the top layer
0.1  ┘
1.5E-03 33.2  0.1  ┐ The first number is the egg diameter in m.
1.5E-03 33.4  0.4  │ This diameter is repeated as many times as it
1.6E-03 33.2  0.4  │ is buoyancy groups, then the next diameter group
1.6E-03 33.4  0.1  ┘ starts. The last column gives the fraction that
                     the diameter/buoyancy group is of the total.
```

In the sample dialogue above the user specifies 50 layers in the datafile and gives a bottomdepth of 100 meters. The depth of each layer, dZ, is then 2 m. The temperature, salinity and concentration data should be straight forward to specify. The relative values of the eddy turbulence coefficient compared to the eddy turbulence computed for the upper layer using (6) is not easy to compute unless the whole water column is homogenous or there is a strong pycnocline and the eggs is above that level. In these cases the values should be set to 1.0, in the case of a strong pycnocline the bottom depth is set to the level of the pycnocline. In other cases only general advice is possible. The reader is adviced to consult the papers of Gargett (1984), Stigebrandt (1985), Omstedt (1985) and Rodi (1980). Gargett gives some guidelines on how to compute K(z) values based upon the stability of the water column below the wind mixed layers in the oceans. In the last three papers more complex models of the turbulence process is presented.

The last part of the datafile specifies the properties of the eggs. It is the salinity at which the egg have neutral buoyancy that is given in the file. The reason is that this is the value measured in laboratory experiments and that the eggs have the internal pressure and temperature as the surrounding water (Sundnes et.al., 1965)

In the next questions the user gives the wind speed at the surface that he wants to use in the simulation, the number of days to run the model and how often he wants the result to be displayed at the screen. The last question the user must deal with is :

Do you want a hardcopy (Y/N) ? Y

If the answer is Y (yes) a figure similar to Fig. 2 is drawn on the pen plotter. The total number of eggs present in the water column is computed every time the figure is displayed to give the user a crude check on the accuracy of the computations. The number of eggs should be constant since they not are subject to death and no new eggs are introduced.

EXAMPLES

The present model have so far not been compared with actual field data except for the staedy state solutions shown by Sundby (1983). For illustrative purposes I have made two examples showing what the eggs properties and different physical conditons leads to quite different vertical concentration profiles of eggs. In both cases the initial concentration profile of eggs is symmetricly distributed around 50 m from 20 to 80 m.

In the first example there is a thermocline at the depth of 50 m and a bottom depth of 100 m. The temperature is 7 ^{0}C above 50 m and 5 ^{0}C below. The salinity is 34 0/00 above 50 m and 35 0/00 below. The wind speed is 5.0 m/s. The turbulence coefficient below the thermocline is supposed to be 1/10 of the value in the wind mixed layer. Two groups of eggs are given one with a diameter of 1.5 mm and neutral buoyancy at 33.0 0/00 and one with a diameter of 1.3 mmm and neutral buoyancy at 36.0 0/00. The result is shown in Fig. 2 after the model has been running for twelve hours. It is seen that the heavy eggs concentrates quite near the bottom, while the lighter eggs with properties near e.g. cod eggs have a pure pelagic distribution. The initial concentration of eggs is given by a dashed line in Fig. 2.



Figure 2. Sample output of the model showing a concentration profile of fish eggs after 12 hours (solid line). The initial concentration is shown with a dashed line.



Figure 3. Sample output of the model showing a concentration profile of copepod eggs after 12 hours (solid line). The initial concentration is shown with a dashed line.

Fig. 3 shows a simulated profile of copepod eggs. Eggs of the copepod Calanus finmarchicus are extremely heavy compared to most fish eggs (Salzen, 1956) and they are seen (Fig. 3) to get a concentration near the bottom in a homogenous 100 m wind mixed water column. The salinity and temperature were specified to be 35 0/00 and 5 ^{0}C respectively. The wind speed were set to 2.0 m/s. Note that despite the fact that the copepod eggs are so heavy compared to fish eggs they stay remarkably high in the water column. The reason is that the eggs have a diameter of only 135 µm. If the eggs had been 500 µm most of the eggs would have been in the lower 5 m of the water column after 12 hours.

DISCUSSION

When the aim is to model the vertical distribution of planktonic eggs over a period of 1-2 days the presented model is well suited. When the model is used in the homogenous wind mixed upper layer the results corresponds well with field observation of fish eggs (Sundby, 1983). At the moment when this report was written there was not available suitable field data to evaluate the model´s ability to describe the vertical concentration of eggs in a stratified water column. It is strongly believed, however, that the model gives a good description as long as the relative strength of the vertical distribution of turbulence is specified correct. Even if this distribution diverge from the real one the model gives reasonable results. The data for the example shown in Fig. 2 was altered so that the relative strength of the turbulence below the thermocline was put to 0.5 intstead of 0.1. There were no drastic changes. The main effects were that the lighter egg group got a deeper distribution while the heavy group raised somewhat.

An important parameter in the model is the Reynold´s number when the description of the raise/sink velocity of eggs switches from equation (1A) to (1B). Davis, 1972, states that equation (1A) is valid upto a Reynold number of 1.0 and in practice even could be extended to Reynold numbers as high as 50.0. I have had no oppurtunity to check the validity of this for pelagic eggs and have adopted the assumption of Sundby, op. cit..

If the object is to describe the vertical distribution of fish eggs over extended time periods several other aspects must be included in the model equations. Below the most important factors that must be included in such a model are described.

The spawning and hatching of eggs must be included and the state of the eggs must be expanded with the development stage of the eggs to allow for the fact that eggs when they develop might change their buoyancy (Solemdal,1967,1971).

If the horizontal gradients is far from homogenous and there is a vertical shear in the currents the presented model is not suitable. The model also lacks the ability to model the effect of shifting wind conditions. More sophisticated models of the effect of wind on the turbulence in the upper layers are then needed. Such a model should also include air-temperature, -humidity and -pressure in addition to wind speed.

A promising example of a better model of the turbulence k-ϵ model (Rodi, 1980, Omstedt, 1985). That model makes it possible to include effects like the deepening of the wind mixed layer and the development of a seasonal thermocline as opposed to the present model where the density profile in the water column is fixed. The benefit of such a model however becomes only manifest when there is spawning over extended time periods.

When the results of the present model are compared to field data I suspect that in areas with strong horizontal and vertical shears in the currents the results will often be disappointing. This is because advection processes here plays an important role. The only way to to make progress in such areas is to define detailed 3D-models.

Despite the shortcomings of the present model it is an useful tool for a researcher who want to gain insight in the vertical mixing process of pelagic fish and plankton eggs. The researcher may experiment with the properties of eggs and physical variables and observe the effects directly as a "movie" on his graphical screen. Such experiments will lead to better insight and inevitably to better field sampling strategies.

REFERENCES

Anon. 1984. GPGS-F user´s guide. 6th edition. Tapir, Trondheim.

Ames, F.A. 1977. Numerical methods for partial differential equations. Second edition. Academic Press.

Davis, J.T. 1972. Turbulence Phenomena. An introduction to the eddy transfer of momentum, mass, and heat, particularly at interfaces. Academic Press.

Fofonoff, N.P. and Millard, R.C. 1983. Algorithms for computation of fundamental properties of seawater. Unesco Tech. Paper in Mar. Sci., No. 44, 53 pp.

Gargett, A.E. 1984. Vertical eddy diffusitivity in the ocean interior. J. Mar. Res. (42):359-393

Krümmel, O. and Ruppin,E. 1905. Wissensch. Meeresuntersuchungen der Kieler Kommission, Vol. 9, p. 29. (Kiel)

Omstedt, A. 1985. Modelling frazil ice and grease ice formation in the upper layers of the ocean. Cold Regions Science and Technology, 11:87-98

Pond, S. and Pickard, G.L. 1983. Introductory dynamical oceanography. 2nd edition. Pergamon Press.

Roache, P.J. 1972. Computational fluid dynamics. Hermosa Publishers. Albaquerque.

Rodi, W. 1980. Turbulence models and their application in hydraulics - a state of the art review. Int. Assoc. Hydrol. Res., Sect. Fundam. Div.2: Exp. Math. Fluid Dyn., Delft, The Netherlands, pp. 1-104.

Stigebrandt, A. 1985. A model for the seasonal pycnocline in rotating systems with application to the Baltic Proper. J. Phys. Oceanogr., 15:1392-1404

Solemdal, P. 1967. The effect of salinity on buoyancy, size and development of flounder eggs. Sarsia, 29:431-442

Solemdal, P. 1971. Prespawning flounders transferred to different salinities and the effects on their eggs. "Vie et Milieu", Troisième Symposium Européen de Biologie Marine. pp 409-423.

Sundby, S. 1983. A one-dimensional model for the vertical distribution of pelagic fish eggs in the mixed layer. Deep Sea Research Vol. 30(6A):645-661.

Sundnes, G., Leivestad, H. and Iversen, O. 1965. Buoyancy determination of eggs from the cod (Gadus morhua L.) J. Cons. perm. int. Explor. Mer, 29:249-252

APPENDIX 1. Sourcecode of the program.

```
C ** (Trond)Model-2:symb **
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C Diffusion/advection model for the vertical distribution of pelagic
C spheres/eggs in the wind mixed layers of the ocean.
C
C Rise/sink velocity of eggs and the eddy diffusivity coefficient in
C the surface layer is computed according to formulaes by Svein Sundby
C given in : C Deep Sea Research 30(6)645:661, 1983.
C
C All units in this program are SI-units i.e. : kg, m and s
C
C The program is written in FORTRAN-77 with a few ND-extensions.  I.e.
:
C
C - The      Do for .... Enddo     construct is used
C - Some variable names are more than 6 characters
C - Both upper and lower case letters are used in the code
C
C Trond Westgård. Institute of Marine Research. Bergen. Norway.
C
C 02.12.1987
C
PROGRAM MODEL2
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C   D E F I N I T I O N S   A N D   D E C L A R A T I O N S   :
C
C-----------------------------------------------------------------------
C              S P A C E   D I M E N S I O N   :
C
C nZmax   = Maximum number of grid steps in the model in Z-direction
      Parameter (nZmax=200)
C nZ      = Number of grid steps in the model in the Z-direction
      Integer nZ
C Z       = Depth in m
      Real Z
C dZ      = Distance between grid points in the Z-direction in m.
      Real dZ
C iZ      = Index of grid point in the Z-direction.
      Integer iZ
C prZ     = Physical range in the Z-direction in m
      Real prZ
C-----------------------------------------------------------------------
C     C O N C E N T R A T I O N   O F   E G G S / S P H E R E S   :
C
C nDmax   = Maximum number of egg diameter groups
      Parameter (nDmax=9)
C nD      = Actual number of egg diameter groups in the run
      Integer nD
C iD      = Index of egg diameter group
      Integer iD
C nBmax   = Maximum number of egg density groups
      Parameter (nBmax=9)
C nB      = Actual number of egg density groups in the run
      Integer nB
```

```
C iB       = Index of egg density group
      Integer iB
C Nall(iZ) = Number of eggs at time zero in each depth level in all
C            egg groups in diameter and buoyancy. Read from file.
      Real Nall(nZmax)
C NFraction = Fraction of egggroup iD/iB at time zero. Read from file.
      Real NFraction
C N(iZ,iD,iB)   = Concentration of eggs pr. m3 in layer iZ,iZ = 1,nZ
C            at timestep iT in diameter group iD and buoyancy group iB
      Real N(0:nZmax+1,1:nDmax,1:nBmax)
C Np1(iZ,iD,iB) = Value of N(iZ,iD,iB) at the next time step
      Real Np1(0:nZmax+1,1:nDmax,1:nBmax)
C  advN     =  Change  in  N(iZ,iD,iB)  during  one  time  step due to
advection
      Real advN
C difN     =  Change  in  N(iZ,iD,iB)  during  one  time  step  due  to
diffusion
      Real difN
C bthN     = Change in N(iZ,iD,iB) during one time step due to
C            local generation (birth).
      Real bthN
C dthN     = Change in N(iZ,iD,iB)  during one time step due to
C            local degeneration (death).
      Real dthN
C totN     = Total amount of N present at time iT*dT
      Real totN
C---------------------------------------------------------------------
C   P R O P E R T I E S   O F   T H E   E G G S / S P H E R E S   :
C
C EggVlc   = Terminal rise/sink velocity of an egg due to buoyancy
C            effects in m/s
      Real EggVlc
C EggSal(iD,iB) = The salinity where the egg has neutral buoyancy
      Real EggSal(1:nDmax,1:nBmax)
C EggDia(iD,iB) = The diameter of an egg in m
      Real EggDia(1:nDmax,1:nBmax)
C EggVmean(iD,iB) = The mean rise/sink velocity of the eggs/spheres
C                   distributed in the iZ,iD,iB space at time zero.
      Real EggVmean(1:nDmax,1:nBmax)
C EggVmall  =  The  mean  rise/sink  of  the  absolute  value  of  the
velocities
C            of all eggs/spheres at time z
      Real EggVmall
C  EggVmax   =  The  max.  rise/sink  velocity  of  the   eggs/spheres
distributed
C            in the iZ,iD,iB space at time zero.
      Real EggVmax
C Reynold = Reynold´s number for the egg´s advection
      Real Reynold
C--------------------------------------------------------------------
C                          T I M E   :
C
C nT       = Number of time steps the model should run
      Integer nT
C dT       = Time step in s
      Real dT
C iT       = Time step number
      Integer iT
C prT      = Total time in seconds the model is allowed to run
      Real prT
```

```
C---------------------------------------------------------------------
C          W A T E R   AND   A T M O S P H E R E   :
C
C Temp(iZ) = Temperature in seawater in degrees Celsius.
C            at depth iZ
      Real Temp(1:nZmax)
C SeaSal(iZ)  = Salinity of seawater in 0/00 at depth iZ
      Real SeaSal(1:nZmax)
C Dens0(SeaSal,Temp) = The density of seawater in kg/m3 at atmospheric
C           pressure. The International Equation of State of seawater.
      Real Dens0
C KinMV   = Kinematic molecular viscosity in m2/s
      Real KinMV
C WindSpd = Windspeed at the surface in m/s
      Real WindSpd
C KV(iZ)  = Vertical turbulence coefficient in m2/s
      Real KV(0:nZmax)
C KVC     = Vertical turbulence coefficient in the upper layers
      Real KVC
C KFraction(iZ) = The relative value of KVC for KV(iZ)
      Real KFraction(nZmax)
C---------------------------------------------------------------------
C                 M I S C E L L A N E O U S   :
C
C KVN     = Approximate numeric diffusion coefficient in m2/s
      Real KVN
C Error   = Percentage error in the transient diffusion coefficient.
      Real Error
C DataFile = Name of file where the salinity, temperature and eggdata
C            at time zero is read from.
      Character DataFile*32
C Screen = The GPGS-F number of the graphical screen used
      Integer Screen
C Plotter = The GPGS-F number of the graphical plotter used
      Integer Plotter
C ShowTime = Number of hours between each graphical output
      Real ShowTime
C IShwTme  = Number of time steps between each output
      Integer IShwTme
C Hours   = The time in hours when there is ShowTime
      Real Hours
C Days    = The numbers of days the user wants to run the model
      Real Days
C c1,c2   = Necessary constants in the calculations.
      Real c1,c2
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C                 I N I T I A L I Z A T I O N   :
C
C Screen and Plotter is the GPGS-F ID of the Tandberg graphical screen
C and the HP-7550 pen plotter.
C
      Screen     = 63
      Plotter    = 5880
      Do for iZ = 1,nZmax,1
         Temp(iZ)   = 0.0
         SeaSal(iZ) = 0.0
      Enddo
      Do for iZ = 0,nZmax,1
```

```
            KV(iZ)       = 0.0
         Enddo
         Do for iZ = 0,nZmax+1,1
            Do for iD = 1,nDmax,1
               Do for iB = 1,nBmax,1
                  N(iZ,iD,iB)    = 0.0
                  Np1(iZ,iD,iB) = 0.0
               Enddo
            Enddo
         Enddo
C
         Write (*,'(''$Give number of eggdiameter groups : '')')
         Read  (*,*) nD
         Write (*,'(''$Give number of eggbuoyancy groups : '')')
         Read  (*,*) nB
         Write (*,'(''$Give number of depth layers       : '')')
         Read  (*,*) nZ
         If (nD.gt.nDmax.or.nB.gt.nBmax.or.nZ.gt.nZmax)  Stop  '  Invalid
valu
        &e of nD, nB or nZ '
C
         Write (*,'(''$Give name of file with data       : '')')
         Read  (*,*) DataFile
         Open (10,file=DataFile,access='rseq',status='old',err=88888)
         Read (10,*,err=99999) (Temp(iZ),iZ=1,nZ)
         Read (10,*,err=99999) (SeaSal(iZ),iZ=1,nZ)
         Read (10,*,err=99999) (Nall(iZ),iZ=1,nZ)
         Read (10,*,err=99999) (KFraction(iZ),iZ=1,nZ)
         Do for iD = 1,nD,1
            Do for iB = 1,nB,1
               Read                                        (10,*,err=99999)
EggDia(iD,iB),EggSal(iD,iB),NFraction
               Do for iZ = 1,nZ,1
                  N(iZ,iD,iB) = Nall(iZ)*NFraction
               Enddo
            Enddo
         Enddo
C
         Write (*,'(''$Give maximum depth in meters      : '')')
         Read  (*,*) prZ
         dZ          = prZ/nZ
C
         Write (*,'(''$Give wind speed in m/s            : '')')
         Read  (*,*) WindSpd
         KVC     = 76.1E-04 + 2.26E-04 * WindSpd * WindSpd
C
         Write (*,'(''$Give the number of days to run the model : '')')
         Read  (*,*) Days
         Write (*,'(''$Give hours between showtimes             : '')')
         Read  (*,*) ShowTime
         prT = Days*24*60*60
C
C
         EggVmax   = -100000.0
         EggVmall = 0.0
         Do for iD = 1,nD,1
            Do for iB = 1,nB,1
               EggVmean(iD,iB) = 0.0
               Do for iZ = 1,nZ,1
                  Call Speed (SeaSal(iZ),Temp(iZ),EggDia(iD,iB),
```

```
     &                          EggSal(iD,iB),EggVlc,KinMV,Reynold)
                 If (Reynold .gt. 5.0) then
                     Write(*,*) 'Warning !! Reynold greater than 5.0 '
                     Pause ' Push CR to continue '
                 Endif
                 If (EggVlc .gt. EggVmax) EggVmax = EggVlc
                 EggVmean(iD,iB) = EggVmean(iD,iB) + EggVlc
                 EggVmall = EggVmall + Abs(EggVlc)
              Enddo
              EggVmean(iD,iB) = EggVmean(iD,iB)/nZ
           Enddo
        Enddo
        EggVmall = EggVmall/(nD*nB*nZ)
C
C  A  correction  of  the  physical  diffusion  is  given  to  get  the
approximate
C steady state solution. A transient numerical diffusion is impossible
C to avoid.
C
        KVN = 0.5*Abs(EggVmall)*dZ
        If ((KVC-KVN) .lt. 0.001) Stop' Numeric diffusion to big '
        Do for iZ = 1,nZ-1,1
           KV(iZ) = KVC - KVN
           If (KV(iZ).lt.0) KV(iZ) = 0.0
           KV(iZ) = KV(iZ)*KFraction(iZ)
        Enddo
C
C Advection/diffusion or pure diffusion. dT is put to a quarter of the
C necessary time step for the upwind differencing method :
C Ideally KVN should be deleted.
C
C        dT = 0.25 * (1./(2.*(KVC-KVN)/(dZ*dZ) + (Abs(EggVmax)/dZ)))
        dT = 0.5 * (1./(2.*(KVC-KVN)/(dZ*dZ) + (Abs(EggVmax)/dZ)))
C
C The approximate effective transient diffusion coefficient KVNT is :
C KVNT = 0.5*Abs(EggVmall)*dZ*(1-c) ; where c = Abs(EggVmall)*dT/dZ
C When KVC is corrected by KVN to get the correct steady state the
C approximate percent error in the effective transient diffusion
C coefficient, Error, is :
C
        Error = 100.*(KVN*Abs(EggVmall)*dT/dZ)/KVC
        Write(*,*) 'Approximate  percentage  error  in  transient  KV  :
',Error
        Pause ' Push CR to continue '
C
        nT  = prT/dT
        IShwTme = Nint((ShowTime*60.*60.)/dT)
        If (IShwTme .eq. 0) IShwTme=1
C
        c1      = dT/dZ
        c2      = dT/(dZ*dZ)
C
C Evaluation with    T I M E    starts   here                 !!!!!!!!!!!
C
        Do for iT = 1, nT, 1
C
C Evaluation over    E G G   G R O U P S   starts   here      !!!!!!!!!!!
C
           Do for iD = 1, nD, 1
              Do for iB = 1, nB, 1
```

```
C
C Evaluation with    S P A C E  in Z-direction strats here !!!!!!!!!!!
C
C Upwind differencing.
C The evaluation of the do-loop direction was also necessary at least
C on the ND-500 computer to get stability.
C
               If (EggVmean(iD,iB) .le. 0) then
C
               Do for iZ = 1,nZ,1
C
                  If (iZ .eq. 1) then
                     EggV1 = 0.0
                     SS = (SeaSal(1)+SeaSal(2))/2
                     TT = (Temp(1)+Temp(2))/2
                     Call Speed (SS,TT,EggDia(iD,iB),EggSal(iD,iB),
     &                           EggV2,KinMV,Reynold)
                  Elseif (iZ .eq. nZ) then
                     SS = (SeaSal(nZ-1)+SeaSal(nZ))/2
                     TT = (Temp(nZ-1)+Temp(nZ))/2
                     Call Speed (SS,TT,EggDia(iD,iB),EggSal(iD,iB),
     &                           EggV1,KinMV,Reynold)
                     EggV2 = 0.0
                  Else
                     SS = (SeaSal(iZ-1)+SeaSal(iZ))/2
                     TT = (Temp(iZ-1)+Temp(iZ))/2
                     Call Speed (SS,TT,EggDia(iD,iB),EggSal(iD,iB),
     &                           EggV1,KinMV,Reynold)
                     SS = (SeaSal(iZ)+SeaSal(iZ+1))/2
                     TT = (Temp(iZ)+Temp(iZ+1))/2
                     Call Speed (SS,TT,EggDia(iD,iB),EggSal(iD,iB),
     &                           EggV2,KinMV,Reynold)
                  Endif
C
                  advN = c1*(EggV2*N(iZ,iD,iB) - EggV1*N(iZ-1,iD,iB))
C
C Central differencing in Z for diffusion (Dag Slagstad pers.comm) :
C
                  difN  = c2*( KV(iZ)    * ( N(iZ+1,iD,iB)-N(iZ,iD,iB))
-
     &                       KV(iZ-1) * ( N(iZ,iD,iB)  -N(iZ-1,iD,iB)) )
                  bthN = +0.0
                  dthN = -0.0
                  Np1(iZ,iD,iB) = N(iZ,iD,iB) +advN +difN +bthN +dthN
               Enddo
            Else
               Do for iZ = nZ,1,-1
C
                  If (iZ .eq. 1) then
                     EggV1 = 0.0
                     Call Speed (SeaSal(2),Temp(2),EggDia(iD,iB),
     &                           EggSal(iD,iB),EggV2,KinMV,Reynold)
                  Elseif (iZ .eq. nZ) then
                     Call        Speed          (SeaSal(nZ-1),Temp(nZ-
1),EggDia(iD,iB),
     &                           EggSal(iD,iB),EggV1,KinMV,Reynold)
                     EggV2 = 0.0
                  Else
                     SS = (SeaSal(iZ-1)+SeaSal(iZ))/2
                     TT = (Temp(iZ-1)+Temp(iZ))/2
```

```
                     Call Speed (SS,TT,EggDia(iD,iB),EggSal(iD,iB),
     &                            EggV1,KinMV,Reynold)
                     SS = (SeaSal(iZ)+SeaSal(iZ+1))/2
                     TT = (Temp(iZ)+Temp(iZ+1))/2
                     Call Speed (SS,TT,EggDia(iD,iB),EggSal(iD,iB),
     &                            EggV2,KinMV,Reynold)
                 Endif
C
                 advN = c1*(EggV2*N(iZ+1,iD,iB) - EggV1*N(iZ,iD,iB))
                 difN = c2*( KV(iZ)    * (  N(iZ+1,iD,iB)-N(iZ,iD,iB))
-
     &                     KV(iZ-1) * ( N(iZ,iD,iB) - N(iZ-1,iD,iB)) )
                 bthN = +0.0
                 dthN = -0.0
                 Np1(iZ,iD,iB) = N(iZ,iD,iB) +advN +difN +bthN +dthN
              Enddo
           Endif
         Enddo
       Enddo
C
C Presentation if ShowTime :
C
        If (iT.eq.1 .or. mod(iT,IShwTme) .eq. 0) then
            Hours=dT*(iT-1.)/3600.
            totN=0.0
            Do for iZ = 1,nZ,1
               Nall(iZ) = 0.0
               Do for iD = 1,nD,1
                  Do for iB = 1,nB,1
                     Nall(iZ) = Nall(iZ) + N(iZ,iD,iB)*dZ
                     totN = totN + N(iZ,iD,iB)*dZ
                  Enddo
               Enddo
            Enddo
            Call                                          ShowResults
(Nall(1),nZ,Hours,totN,prZ,Screen,Plotter)
        Endif
C
C Makes ready for a new time step :
C
        Do for iZ=1,nZ
           Do for iD=1,nD
              Do for iB=1,nB
                 N(iZ,iD,iB)=Np1(iZ,iD,iB)
              Enddo
           Enddo
        Enddo
C
      Enddo
C
      Stop ' Normal termination of the session. Good Bye !!! '
88888 Stop ' Error when opening datafile '
99999 Stop ' Error when reading on given datafile '
C
      End
C
C ** ShowResults **
C
      Subroutine ShowResults (Array,NoInArray,Hours,CheckSum,Length,
     &                        Screen,Plotter)
```

```
C
C A routine to show the resulting egg/sphere profile to a graphic
C medium.
C
      Logical FirstTime, HardCopyDone
      Data FirstTime /.true./
      Character Answer*1
      Integer Screen, Plotter, I, NoInArray
      Real Array(1:NoInArray),Hours,CheckSum,Length,Z
      Real Xmin,Xmax,Rmin,Rmax,Xstep
C
C Initialization of the plot package GPGS-F form NORSIGD :
C
      If (FirstTime) then
          Call Gpgs
          Call DefPag (0.10,1.00,1)
          Call DefPag (0.10,0.90,2)
          Call Cfprop (0)
          Call Cfont  (1)
          Call Nitdev (Screen)
          Call Nitdev (Plotter)
          Delta = Length/NoInArray
          FirstTime = .false.
      Endif
C
C Plots the vertical concentration profile of eggs/spheres :
C
      HardCopyDone = .false.
      Call Seldev (Screen)
   10 Call Bgnpic (1)
C        X-axis :
         Call Mimatb (Array(1),NoInArray,1,Rmin,Rmax)
         Call NicLab (Rmin,Rmax,5,Xmin,Xmax,Xstep)
         Xmin=0.
         Call DefAx  (Xmin,Xmax,Xstep,1)
C        Y-axis :
         Call DefAx  (Length,0.0,-25.0,2)
C        The drawing :
         Call Axis   (1,1)
         Call AxaLab
         Call AxTit  ("Concentration (no./m3)*.")
         Call Axis   (1,2)
         Call Crotad (-90.0)
         Call AxaLab
         Call Crotad (0.0)
         Call AxTit  ("Depth (m)*.")
         Call PloLin (Xmin,0.,0)
         Call PloLin (Xmax,0.,1)
         Call PloLin (Xmax,Length,1)
         Call PloLin (Array(1),0.,0)
         Z = 0.
         Do for I = 1,NoInArray,1
            Z =  Z + Delta
            Call PloLin (Array(I),Z,1)
            If (I.lt.NoInArray) Call PloLin (Array(I+1),Z,1)
         Enddo
         Call Line   (0.11,0.91,0)
         Call Charc  ('Time in hours  : *.')
         Call Charf  (Hours,6,2)
         Call Line   (0.50,0.91,0)
```

```
          Call Charc  ('Number of eggs : *.')
          Call Chare  (CheckSum,10,3)
       Call Endpic
C
       If (.not. HardCopyDone) then
           Write (*,'(''$Do you want a hardcopy (Y/N) ? '')')
           Read  (*,*) Answer
           If (Answer .eq. 'N' .or. Answer .eq. 'n') then
               Call Clrdev (Screen,0)
               Return
           Elseif (Answer .eq. 'Y' .or. Answer .eq. 'y') then
               HardCopyDone = .true.
               Call Seldev (Plotter)
               Goto 10
           Endif
       Elseif (HardCopyDone) then
           Call Clrdev (Screen,0)
           Call Clrdev (Plotter,0)
           Return
       Endif
       End
C
C ** Speed **
C
       Subroutine                                                    Speed
(SeaSal,Temp,EggDia,EggSal,EggVlc,KinMV,Reynold)
C
C The equations are from S.Sundby 1983
C
C Trond Westgård, Institute of Marine Research, Bergen, Norway
C
C 10.12.1987
C
C SeaSal  = Salinity of seawater in 0/00
C Temp    = Temperature of the water in OC
C EggDia  = Diameter of the sphere in m
C EggSal  = Salinity where the sphere has zero buoyancy in 0/00
C EggVlc  = Rise/sink velocity of the sphere in m/s
C KinMV   = Kinematic molecular viscosity in m2/s
C Reynold = Reynolds number for the advection of the sphere
C Agrav   = Acceleration of gravity in m/s2
C
       Real SeaSal,Temp,EggDia,EggSal,EggVlc
       Real KinMV,Dmax,DelRho,Reynold,Agrav
       Data Agrav /9.81/
C
       DelRho = Dens0 (SeaSal,Temp) - Dens0 (EggSal,Temp)
C
       KinMV  = 1.854E-06*EXP(-0.02783*Temp)
C
C Stoke's equation applies up to Reynold's number equal to 0.5.
C The corresponding diameter Dmax is derived using equation (1) in
C Sundby 1983.(Multiply by d and divide by "ny" and put eq to 0.5)
C
       Dmax   = (9.0E03*(KinMV**2)/(ABS(DelRho)*Agrav))**0.3333333
C
       If (EggDia .lt. Dmax) then
C
C Stoke's equation applies :
C
```

```
         EggVlc = (1.E-03/18.)*Agrav*(EggDia**2)*DelRho/KinMV
C
      Else
C
C DallaValle equation applies (up to Reynold numbers equal to 5.) :
C
         EggVlc = 8.825E-03 * (EggDia - 0.4*Dmax) * SIGN(1.,DelRho)*
     &               (ABS(DelRho)**(2./3.)) *
     &               (KinMV**(-1./3.))
C
      Endif
C
      Reynold = ABS(EggVlc)*EggDia/KinMV
      Return
      End
C
C ** Dens0 **
C
      Real Function Dens0 (S,T)
C
C The function gives the density of seawater at one atmosphere
C pressure as given in :
C
C N.P. Fofonoff and R.C. Millard Jr.,1983,
C Unesco technical papers in marine science no. 44.
C
C S   = Salinity in promille of the seawater
C T   = Temperature of the seawater in degrees Celsius
C
      Real S,T,SIG,R1,R2,R3,R4,DR350
      Data R4/4.8314E-04/,DR350/28.106331/
C Pure water density at atmospheric pressure
C Bigg P.H. (1967) BR. J. Applied Physics pp.:521-537
C
      R1 = ((((6.536332E-09*T-1.120083E-06)*T+1.001685E-04)*T
     &          -9.095290E-03)*T+6.793952E-02)*T-28.263737
C
C Seawater density at atmospheric pressure
C coefficients involving salinity :
C
      R2 = (((5.3875E-09*T-8.2467E-07)*T+7.6438E-05)*T-4.0899E-03)*T
     &         +8.24493E-01
C
      R3 = (-1.6546E-06*T+1.0227E-04)*T-5.72466E-03
C
C International one-atmosphere equation of state of seawater :
C
      SIG = R1 + (R4*S + R3*SQRT(S) + R2)*S
      Dens0 = SIG + DR350 + 1000.
      Return
      End
```

Oversikt over tidligere utkomne rapporter.

1987

Nr. 1 P. Solemdal og P. Bratland: Klekkeforløp for lodde i Varangerfjorden 1986.

Nr. 2 T. Haug og S. Sundby: Kveitelarver og miljø. Undersøkelser på gytefeltene ved Sørøya.

Nr. 3 H. Bjørke, K. Hansen og S. Sundby: Postlarveundersøkelser i 1986.

Nr. 4 H. Bjørke, K. Hansen og W. Melle: Sildeklekking og seigyting på Møre 1986.

Nr. 5 H. Bjørke and S. Sundby: Abundance indices for the Arcto-Norwegian cod in 1979-1986 based on larvae investigations.

Nr. 6 P. Fossum: Sult under larvestadiet - en viktig rekrutteringsmekanisme?

Nr. 7 P. Fossum og S. Tuene: Loddelarveundersøkelsene 1987.

Nr. 8 P. Fossum, H. Bjørke and R. Sætre: Studies on herring larvae off western Norway in 1986.

Nr. 9 K. Nedreaas and O.M. Smestad: O-group saithe and herring off the Norwegian coast in 1986 and 1987.

Nr. 10 P. Solemdal: Gytefelt og gyteperiode hos norsk-arktisk hyse.

Nr. 11 B. Ellertsen: Kopepodnauplier på Møre våren 1986 - næringstilbudet til sildelarver.

Nr. 12 H. Bjørke, P. Fossum, K. Nedreaas og R. Sætre: Yngelundersøkelser - 1985.

Nr. 13 Faglig profil og aktivitetene i 1986-87.

Denne rapportserien har begrenset distribusjon. Opplysninger om programmet og rapportene kan rettes til

Programledelsen for HELP
Fiskeridirektoratets Havforskningsinstitutt
Postboks 1870
5024 Bergen

1988

Nr. 14 H. Bjørke, K. Hansen, M. Johannessen og S. Sundby: Postlarveundersøkelser - juni/juli 1987.

Nr. 15 H. Bjørke: Sildeklekking på Møre i 1986-87.

Nr. 16 H. Bjørke, K. Bakkeplass og K. Hansen: Forekomster av fiskeegg fra Stad til Gimsøy i februar-april 1987.